/*
 * Copyright (c) 2016 Appolica Ltd.
 *
 * Licensed under the Apache License, Version 2.0 (the "License"); you may not use this file
 * except in compliance with the License.
 *
 * You may obtain a copy of the License at http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software distributed under
 * the License is distributed on an "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND,
 * either express or implied. See the License for the specific language governing permissions and
 * limitations under the License.
 */

apply plugin: 'com.android.library'
apply plugin: 'maven-publish'
apply plugin: 'signing'

apply from: 'versioning.gradle'

android {
    compileSdkVersion 30

    defaultConfig {
        minSdkVersion 15
        targetSdkVersion 30
        versionCode 7
        versionName getVersionName()
    }
    buildTypes {
        release {
            minifyEnabled false
            proguardFiles getDefaultProguardFile('proguard-android.txt'), 'proguard-rules.pro'
        }
    }
    compileOptions {
        sourceCompatibility JavaVersion.VERSION_1_7
        targetCompatibility JavaVersion.VERSION_1_7
    }
}

dependencies {
    implementation fileTree(dir: 'libs', include: ['*.jar'])
    implementation "androidx.appcompat:appcompat:1.3.0"
    implementation 'com.google.android.gms:play-services-maps:17.0.1'
}

afterEvaluate {
    publishing {
        publications {
            aar(MavenPublication) {
                from components.release
                artifactId = 'interactive-info-window-android'
                groupId = "com.appolica"
                version = getVersionName()
            }
        }
    }
}

publishing {
    repositories {
        maven {
            url = 'https://s01.oss.sonatype.org/content/repositories/releases/'
//            url = 'https://s01.oss.sonatype.org/content/repositories/snapshots/'   // Repository for snapshot versions of library.
            credentials {
                username = ossrhUsername
                password = ossrhPassword
            }
        }
    }
}

signing {
    def signingKeyId = findProperty("signing.keyId")
    def signingPassword = findProperty("signing.password")
    def signingKey = findProperty("signing.secretKeyRingFile")
    useInMemoryPgpKeys(signingKeyId, signingKey, signingPassword)
}